# Analysis GK Abituraufgabe Bayern 2009

Dokumentnummer: DX1132

Die erste Teilaufgabe der zentralen Aufgabenstellung für das
Abitur 2009 ist zu lösen.
Quellen: http://ne.lo-net2.de/selbstlernmaterial/m/abi/BY/mathgk09_A.pdf

```
--> kill(all);
(%o0) done
```

Figure 1: Grundinformation

Die Abbildung zeigt den Graphen $G_f$ einer ganzrationalen Funktion f dritten Grades mit dem Definitionsbereich $D_f = \mathbb{R}$. Die in der Abbildung angegebenen Punkte $P(1|3)$, $N_1(0|0)$, $N_2(2|0)$ und $N_3(4|0)$ sind Punkte von $G_f$.

4 Punkte sind gegeben

```
--> P:[1,3];N1:[0,0];N2:[2,0];N3:[4,0];
(%o1) [1,3]
(%o2) [0,0]
(%o3) [2,0]
(%o4) [4,0]
```

Die Punkte werden in Koordinaten zerlegt

```
--> x1:P[1];y1:P[2];
x2:N1[1];y2:N1[2];
x3:N2[1];y3:N2[2];
x4:N3[1];y4:N3[2];
(%o5) 1
(%o6) 3
(%o7) 0
(%o8) 0
(%o9) 2
(%o10) 0
(%o11) 4
(%o12) 0
```

Figure 2: Der Graph



Figure 3: Die Funktion kann man leicht erraten, wie mit Geogebra gezeigt wird




Figure 4: Bestimmung des Polynoms

Geben Sie den Funktionsterm von f in der Form $f(x) = a(x-b)(x-c)(x-d)$ an, indem Sie passende Werte für $a, b, c, d \in \mathbb{R}$ ermitteln. Zeigen Sie, dass sich dieser in der Form $f(x) = x^3 - 6x^2 + 8x$ schreiben lässt.

--> x*(x-2)*(x-4),expand /* die erratene Lösung */;

(%o13) $x^3 - 6x^2 + 8x$

--> g(x,y):=y=a*(x-b)*(x-c)*(x-d);

(%o14) $g(x, y) := y = a(x-b)(x-c)(x-d)$

Anmerkung: b,c und d ergeben sich aus den Nullstellen.

--> g1:g(x1,y1);g2:g(x2,y2);g3:g(x3,y3);g4:g(x4,y4) /* Einsetzen von 4 Punkten */;

(%o15) $3=a(1-b)(1-c)(1-d)$

(%o16) $0=-a\,b\,c\,d$

(%o17) $0=a(2-b)(2-c)(2-d)$

(%o18) $0=a(4-b)(4-c)(4-d)$

--> l:solve([g1,g2,g3,g4],[a,b,c,d]);

(%o19) $[[a=1,b=4,c=2,d=0],[a=1,b=4,c=0,d=2],[a=1,b=2,c=4,d=0],[a=1,b=0,c=4,d=2],[a=1,b=2,c=0,d=4],[a=1,b=0,c=2,d=4]]$

Es ergeben sich mehrere Lösungen, weil die Nullstellen in beliebiger Reihenfolge genommen
werden können. Der klassische Ansatz wird im Folgenden betrachtet:

--> g(x,y):=y=a*x**3+b*x**2+c*x+d;

(%o20) $g(x,y):=y=a\,x^3+b\,x^2+c\,x+d$

--> g1:g(x1,y1);g2:g(x2,y2);g3:g(x3,y3);g4:g(x4,y4);

(%o21) $3=d+c+b+a$

(%o22) $0=d$

(%o23) $0=d+2\,c+4\,b+8\,a$

(%o24) $0=d+4\,c+16\,b+64\,a$

--> l:solve([g1,g2,g3,g4],[a,b,c,d]);

(%o25) $[[a=1,b=-6,c=8,d=0]]$

--> t:rhs(g(x,y)),l[1];

(%o26) $x^3-6\,x^2+8\,x$

--> f(x):=''t;

(%o27) $f(x):=x^3-6\,x^2+8\,x$

Figure 5: Wendepunkt und Wendetangente

Weisen Sie nach, dass $N_2$ Wendepunkt von $G_f$ ist, und ermitteln Sie die Gleichung der zugehörigen Wendetangente.

--> ab2:diff(f(x),x,2);

(%o28) $6x-12$

--> l:solve(ab2=0,x);

(%o29) $[x=2]$

--> f(x),l;

(%o30) $0$

N2 ist ein Wendepunkt

--> ab:diff(f(x),x);

(%o31) $3x^2-12x+8$

--> k:ab,x=x3;

(%o32) $-4$

Die Steigung der Wendetangente ist -4

--> g:y=k*x+d,x=x3,y=y3;

(%o33) $0=d-8$

--> l:solve(g,d);

(%o34) $[d=8]$

--> tg:y=k*x+d,l;

(%o35) $y=8-4x$

Das ist die Gleichung der Wendetangente

Figure 6: Innenwinkel eines Dreiecks

Die Wendetangente schließt mit den Koordinatenachsen ein Dreieck ein. Bestimmen Sie die Innenwinkel dieses Dreiecks.

Figure 7: Geogebra verschafft rasch einen Überlick




--> tg;

(%o36) $y = 8 - 4\,x$

--> l1:tg,x=0;

(%o37) $y = 8$

--> ak:y,l1 /* Abschnitt auf der y-Achse */;

(%o38) 8

--> l2:tg,y=0;

(%o39) $0 = 8 - 4\,x$

--> l3:solve(l2,x);

(%o40) $[x = 2]$

--> gk:x,l3 /* Abschnitt auf der x-Achse */;

(%o41) 2

--> h:sqrt(ak**2+gk**2),numer;h:floor(h*100+0.5)/100.0 /*Lehrsatz von Pythagoras */;

(%o42) 8.246211251235321

(%o43) 8.25

--> w1:asin(gk/h);w1:w1*180/%pi,numer;w1:floor(w1*100+0.5)/100.0 /* Winkel an der y-Achse */;

(%o44) 0.24486385420562

(%o45) 14.02966540128882

(%o46) 14.03

--> w2:90-w1 /* Winkel an der x-Achse */;

(%o47) 75.97

Figure 8: Integralfunktion

Betrachtet wird nun die Integralfunktion $F: x \mapsto \int_0^x f(t)dt$ mit $D_F = \mathbb{R}$.

--> kill(t);

(%o48) *done*

--> f(x);

(%o49) $x^3 - 6x^2 + 8x$

--> f(t);

(%o50) $t^3 - 6t^2 + 8t$

--> F(x):='integrate(f(t),t,0,x);

(%o51) $F(x) := \int_0^x f(t)\,dt$

--> F(x);

(%o55) $\frac{x^4 - 8x^3 + 16x^2}{4}$

Figure 9: Flächenbestimmung

Berechnen Sie $F(4)$. Was folgt daraus für die beiden Flächenstücke, die der Graph $G_f$ mit der x-Achse im I. und im IV. Quadranten einschließt? Begründen Sie Ihre Antwort. Bestimmen Sie nun die Summe der Inhalte dieser beiden Flächenstücke.

```
--> F(4) /* von 0 bis 2 hebt sich mit 2 bis 4 auf,
       siehe Graph */;
(%o54) 0
```

```
--> Flaeche:2*F(2);
(%o53) 8
```

Figure 10: Graph der Integralfunktion

Einer der drei abgebildeten Graphen I, II oder III stellt den Graphen von F dar. Geben Sie an, welcher dies ist, und begründen Sie Ihre Antwort, indem Sie erklären, warum die beiden anderen Graphen nicht in Betracht kommen.




Es muss der Graph II sein ( Polynom 4. Grades), III hat einen höheren Grad, I kommt nicht in Frage, weil es keine negativen Werte geben kann (den Fall, dass F(x)=0 ist, hatten wir schon - alle anderen Werte müssen positiv sein (siehe Grafik).

--> wxplot2d([F(x)], [x,-5,5], [y,0,5])$

plot2d: some values were clipped.

(%t57) 


Figure 11: Unterscheidung Integralfunktion - Stammfunktion

> Bekanntlich ist jede Integralfunktion der Funktion f auch Stammfunktion von f. Begründen Sie, dass jede Integralfunktion mindestens eine Nullstelle hat. Geben Sie den Term einer Stammfunktion von f an, die keine Integralfunktion von f ist.

Hier ist es wichtig, die Aufgabenstellung ganz genau zu lesen, und zwar den zweiten Satz: jede Integralfunktion hat mindestens eine Nullstelle. Wenn die Untergrenze und die Obergrenze gleich ist, dann ist der Wert NULL.

--> f(x);

(%o59) $x^3-6x^2+8x$

--> F(x);

(%o60) $\frac{x^4-8x^3+16x^2}{4}$

--> f(t);

(%o62) $t^3-6t^2+8t$

--> G(x):='integrate(f(t),t,-3,x);

(%o64) $G(x):=\int_{-3}^{x} f(t)\,dt$

--> G(x);

(%o65) $\frac{x^4-8x^3+16x^2}{4}-\frac{441}{4}$

## Stammfunktion

--> S:integrate(f(x),x);

(%o67) $\frac{x^4}{4}-2x^3+4x^2$

--> S:S+2 /* das ist eine Stammfunktion, aber keine Integralfunktion */;

(%o76) $\frac{x^4}{4}-2x^3+4x^2+4$

--> wxplot2d([S], [x,-5,5], [y,0,7])$

plot2d: some values were clipped.

(%t75)




Created with wxMaxima.